
News 
Spike2 6.02 nearing release 

This version includes the facility to export Spike2 files directly to MATLAB and also 
contains script commands for MATLAB communication. See the Spike2 article below for 
further details. 

Other new features include spike collision analysis to separate pairs of spikes by 
comparing the recorded waveform with pairs of templated spike shapes, interactive 
selection, and re-coding of spikes when displayed in Overdraw WaveMark mode and the 
option of display of event time on the Z-axis in the clustering dialog. 

 

Meetings and events 
Thank you to all those who visited our booths during the Society for Neuroscience 
meeting in Atlanta 15th – 18th October and the American Heart Association Scientific 
Sessions meeting in Chicago 12th – 14th November. 

 

IMN Workshop on Spinal Cord Injury 2006 
UCL Institute of Neurology, London, UK 
December 8th 2006 

Biophysical Society Annual Meeting 2007 
Baltimore, Maryland, USA 
March 3rd – 7th 2007 

 

  Latest versions of Spike2 and Signal 
The latest updates for Spike2 and Signal are available from the CED downloads page, or 
by clicking on the links in the table below. Demonstration versions of the latest software 
are also available.  
 

 Spike2 downloads Signal downloads 
Spike2 version 6.01  Signal version 3.07 

 Spike2 version 5.16  Signal version 2.16 

 Spike2 v6 demo  Signal demo 

A full list of the new features and changes in the latest software versions is available from 
the website. 

Spike2   

 

Q. It would be very useful to be able to export Spike2 files to MATLAB for further 
analysis. Do you have any plans to allow this in future versions? 

A. The soon-to-be released version 6.02 of Spike2 includes support for external 
exporters. These allow you to save data, result and XY view data to other file 
formats. The first available exporter writes Spike2 data to MATLAB files. 

Once MATLAB support has been selected during the Spike2 installation, you can 
export Spike2 data as .mat files from the Export as... function in the Spike2 File 
menu. To export XY or result view data, select Save as... and the .mat file type. 
Associated dialogs control the time range of data, the channel types to include 
and the required format, depending on the type of Spike2 file you wish to export. 

 
Export settings dialog for data files 

 

As well as support for the export of Spike2 files, there are new script commands 
to move values between Spike2 and MATLAB variables and execute arbitrary 
commands in MATLAB. Unlike the file export function, these script commands 
can only be used on systems that have a licensed copy of MATLAB installed. 

Full documentation on these new functions can be found in the Spike6\Export 
folder after installation of Spike2 6.02. 

MATLAB is a registered trademark.  

 

  

 

Q. I use multiple states in Signal to output a series of different pulse paradigms and 
control the ordering of these states during an experiment using protocols, but 
have found that only being able to specify 10 steps per protocol is not enough for 
me. I am told that you can create multiple protocols for control of state ordering 
but these then have to be selected by hand during an experiment from the drop-
down list in the states toolbar. Is there any way to automate the running of these 
protocols during an experiment? 

A. The ordering of multiple states in Signal during an experiment can be controlled 
manually from the states toolbar, set to run in a numeric or random order, or 
controlled by protocols, where the user specifies the order and number of repeats 
for the multiple states when sampling. The latest version of Signal includes 
enhancements to the protocol dialog that make it easier to setup multiple 
protocols and allow the user to link them to run sequentially during an experiment. 
There are also new options for creating toolbar buttons for protocols and 
controlling the number of protocol repeats. 

 
Enhanced protocol dialog in Signal 

Scripts: Spike2  
Q. I am recording multimedia files alongside neuronal response data. During periods 

of low response activity, I am making use of the slow frame rate feature in the 
Spike2 video application to keep my file sizes manageable. Is there any way to 
automatically switch to the slow frame rate when activity drops below a certain 
level on the specified channel? 

A. The attached script, VideoRate.s2s, can be used to monitor an event or 
WaveMark channel and switch between slow frame rate and the normal frame 
rate depending on the level of activity in the nominated channel. You can specify 
the minimum interval between events that signals the start of a period of low 
activity. 

 

Scripts: Signal  

Q. I would like to perform a sequential method of leak subtraction that doesn’t seem 
to be supported by the leak subtraction option in the Analysis menu. In my 
example I have 30 frames: the first 15 recorded using a control solution, with the 
next 15 in a blocker solution. What I would like is for the second set of frames to 
be subtracted from the first set of frames in sequence, i.e. frame 16 subtracted 
from frame 1, frame 17 subtracted from frame 2… etc. 

A. The attached script, SubFrameSeries.sgs, allows you to subtract one series 
of frames from another. The script expects a standard voltage clamp setup, i.e. 
one channel holding the stimulus and one channel holding the response data and 

Right-click the script icon 
and save to disk. 

If you have any problems 
opening the embedded 
scripts in this newsletter 
please let us know.  

If you have any problems 
opening the embedded 
scripts in this newsletter 

'$VideoRate.s2s|Spike2 script to set video capture frame rate based on occurence of spiking
 
var eChan%;					'Event channel for triggering video	
var del := 3;				'Interval since last spike before turning off video fast frame rate
var data%;					'Handle of data file
var mm%;																	
var sFRate := 0.02;			'Slow frame rate
var fFRate := 30;			'Fast frame rate
var fFlag%;					'Flag for fast or slow frame rate	
var lSpike$;				'String to be printed to toolbar showing time from last spike
var gap;					'Time from last spike detected
var TMChan% := 30;			'Text mark channel 
var code%[4];																
var gFloat%[20];              'global for floating window states


HideAll();          'Hid windows and toolbars

ToolbarSet(1, "Quit", Quit%);       'Quits the script
ToolbarSet(3, "New file", New%);    'Opens a new file
ToolbarSet(5, "Start", Start%);     'Start sampling
ToolbarEnable(5,0);                 'Disable Start button
mm% := ProgRun("C:\\Spike5\\s2video.exe",1,70,70,100,100);  'Open s2video app
Yield(1);       'Give s2video time to open
Toolbar("Video capture control",1023);      'Set toolbar


Func New%()
SampleTextMark(100);        'Create a TextMark channel
data% := FileNew(0,1);      'Open a new file ready for sampling
DrawMode(TMChan%,15,2);     'Draw TextMark channel in state mode
Window(0,0,100,100);        'Maximise sampling window
FrontView(data%);           'Make sure data file is at the front
DlgCreate("Settings for automated frame rate");     'Dialog to set channel to monitor and level of activity
DlgChan(1,"Channel to monitor",2+16);
DlgReal(2,"Min. Interval between events (s)",0,100);
DlgShow(eChan%,del);

ToolbarEnable(5,1);     'Enable sample start button
ToolbarEnable(3,0);     'Disable Quit button
return 1;
end;


Func Start%()
View(data%);        'Make sure we are viewing the data file
SampleStart();      'Start sampling
MMRate(sFRate);     'Set frame rate for video capture
fFlag% := 0;        'Set flag for normal rate
ToolbarSet(0,"Idle",Idle%);     'Create idle routine to monitor channel
ToolbarClear(5);        'Clear Start button
ToolbarSet(5,"Stop",Stop%);     'Set Stop button
return 1;
end;


Func Stop%()
View(data%);        'Make sure we are viewing the data file
SampleStop();       'Stop sampling
ToolbarClear(5);    'Clear Stop button
ToolbarSet(5, "Start",Start%);      'Set Start button
ToolbarEnable(5,0); 'Disable Start button
ToolbarEnable(3,1); 'Enable New File button
return 1;
end;


Func Idle%()
View(data%);
if LastTime(eChan%,MaxTime()) > 0 then  'Check for events on monitored channel
	gap := MaxTime() - LastTime(eChan%,MaxTime());      'Check that gap between current and last event is less than min. interval
	if gap <= del then      'if so..
		if fFlag% = 0 then
			MMRate(fFRate);     'Set normal frame rate 
			fFlag% := 1;        'Set flag for normal frame rate
			code%[0] := 1;      'Set TextMark
			SampleText("Video running",MaxTime(),code%[]);  'and text
		endif;
	else
		if fFlag% = 1 then      'If flag is already set for normal frame rate
			MMRate(sFRate);     'Set slow frame rate
			fFlag% := 0;        'Set flag for slow frame rate
			code%[0] := 00;     'Reset TextMark
			SampleText("",MaxTime(),code%[]);   'with blank text
		endif;
	endif;
	lSpike$ := Print$("Time since last spike = %ds",Trunc(gap));    'Keep a count of time elapsed between events
	ToolbarText(lSpike$);       'Display count in toolbar
endif;
return 1;
end;

Func Quit%();
RestoreAll();   'Restore all windows and toolbars before quitting the script
return 0;
end;

proc HideAll()
var i%;
gFloat%[0] := App(-2);        'number of windows
for i% := 1 to gFloat%[0] do  'hide all windows and save state
   gFloat%[i%] := View(App(i%)).WindowVisible(0);
   next;
end

proc RestoreAll()
var i%;
for i% := 1 to gFloat%[0] do  'restore hidden windows
   View(App(i%)).WindowVisible(gFloat%[i%]);
   next;
end



VideoRate.s2s


'$SubFrameSeries|Script to subtract one series of frames from another in the same file. 
' The results are displayed in a new memory view.

var dataFile%;		'Handle of data file
var destFrame$;		'Destination frames
var srceFrame$;		'Source frames
var dest%[100];		'Array to hold dest frames
var srce%[100];		'Array to hold source frames
var ok%;			'OK for dialog
var noPoints%;		'No. of data points per frame
var subView%;		'New memory view for results
var yUnits$;		'Y-axis units
var yTitle$;		'Y-axis title
var c%;				'Counter for loop

if ViewKind() = 0 then		'Check selected view holds data
	dataFile%:= View();				'Get the file handle
else
	Message("No data file selected! Open a data file to continue");		'If current view is not a data file, warn user and prompt
	dataFile%:= FileOpen("",0);		'Select data file
endif;

Window(0,0,100,100);		'Position window
Frame(1);					'Go to first frame	
FrontView(dataFile%);			'Bring to front
Optimise(-1);				'Optimise the data
yUnits$:= ChanUnits$(1);	'Get Y-axis units
yTitle$:= ChanTitle$(1);	'Get Y-axis title
noPoints%:= 1/BinSize(1)*(MaxTime()-MinTime());		'Get no. of points for the frame
	

DlgCreate("Frames for Subtraction");			'Settings
DlgString(1,"Control (dest) frames",20);		'Frames to subtract from (destination)
DlgString(2,"Blocker (source) frames",20);		'Frames to subtract	(source)
ok%:= DlgShow(destFrame$,srceFrame$); 			

FrameList(dest%[],destFrame$);		'Get list of dest frames
FrameList(srce%[],srceFrame$);		'Get list of source frames

DoSub%();	'Function to do the work.

Func DoSub%();
var resultA[noPoints%];		'Array to hold data copied from channel 1
var resultV[noPoints%];

subView%:= SetMemory(2,noPoints%,View(dataFile%).BinSize(1),View(dataFile%).MinTime(),0,0,0,"Subtract results","s",yUnits$,"",yTitle$);		'New view to hold results

for c%:=1 to dest%[0] do
	FrontView(dataFile%);		'make sure data file is current view
	Frame(dest%[c%]);			'Step to destination frame
	ArrConst(resultA[],View(dataFile%,1).[]);		'Copy channel data
	ArrConst(resultV[],View(datafile%,2).[]);

	Frame(srce%[c%]);			'Step to source frame
	ArrSub(resultA[],View(dataFile%,1).[]);			'Subtract source channel data from destination
	ArrSub(resultV[],View(dataFile%,2).[]);

	View(subView%).Frame(c%);	'Step to current frame in the result view
	ArrConst(View(subView%,1).[],resultA[]);	'Copy the results to the view
	ArrConst(View(subView%,2).[],resultV[]);
	if c% < dest%[0] then			'If more frames to copy
	View(subView%).AppendFrame();	'Create a new frame in the result view
	endif;
next;

FrontView(dataFile%);	'Make data file current
Window(0,0,50,100);		'Tile to the left of the screen
Frame(1);			'Reset back to frame 1

View(subView%).Window(50,0,100,100);	'Position results to the right
View(subView%).WindowVisible(1);		'Make the result view visible
View(subView%).Frame(1);			'Set to frame 1
View(subView%).Optimise(1);			'Optimise

return 0;
end;


SubFrameSeries.sgs



then lets the user enter a series of destination frames (the ones to subtract data 
from) and a series of source frames (the ones to subtract). You can enter lists of 
frames in the normal format for example, 1,2,5 or 1..15, as used in most dialogs. 

Did you know…? 
The channel process dialog in the latest version of Spike2 now allows you to copy a set 
of applied processes to all selected channels in a data file. 

Recent questions  
Q. I am recording EMG data for force feedback using Spike2 and would like to know 

if there is any way to have the incoming data displayed in the centre of the x-axis 
time range during a recording. This is important for me as it is necessary for the 
subject to be able to see ‘ahead’ of the current time to clearly know when to start 
responding. I am asking the subject to follow a shape drawn on the screen.  

A. The attached example script, SampleWindow.s2s, utilises an idle routine to 
display the incoming data in the middle of the X-axis time range. The user can 
specify the length of the displayed time range before sampling. Although the 
script works well when displaying a few channels at relatively low sample rates, 
such as EMG recordings, the smoothness of the scrolling display will be affected 
by an increase in the number of channels and the sample rate. The exact effects 
will depend on your PC’s hardware configuration and graphics capabilities. 

 

User group      

If you have any comments about the newsletter format and content, or wish to un-
subscribe from the mailing list for this newsletter, please notify sales@ced.co.uk

. 

'$SampleWindow|Keeps incoming data in the middle of the X-axis display
'It contains toolbar to open a new data file for sampling and
'start and stop sampling. Keeps incoming data positioned in the 
'middle of the X-axis time range so that the subject can see 'ahead' of current
'time.  Display gets markedly slower the more channels and greater sampling rates used.

'Hides/restores unwanted windows. Manages data files (one file at a time).

var data%;								'Handle of new data file (Spike2 sets it to 0)
var sTime;								'The last time we looked at the idle routine
var range:=10;
var ok%;
var off;

HideAll();                       'Hide all unwanted bars
ToolbarVisible(1);					'Make toolbar visible always
New%();									'Set up new sampling window
DoToolbar();							'Do the toolbar
RestoreAll();                    'Restore all bars
Halt;

func New%()								'New sampling window
if data%>0 then						'If there is already a data view open...
  View(data%);							'...close it
  FileClose();                   'No point setting data% to 0...
endif;

data%:=FileNew(0,1);					'Open a new data file for sampling based on current sampling config.
if data%<0 then Message("Unable to open a data file for sampling!\nCheck 1401 is connected and switched on");
       RestoreAll();
    Halt() endif;

DrawMode(-1,2);						'Set event draw mode to lines
Window(0,0,100,100);					'Make data window in top bit of screen

DlgCreate("Set X axis time range");     'Set display range
DlgReal(1,"Set time range to display",1,100);
ok%:= DlgShow(range);

if ok% > 0 then
    XRange(0,range);
    off:= range/2;
else
    XRange(0,10);
    off:= 5;
endif;

FrontView(data%);						'Bring the data view to the front
ToolbarEnable(3,0);					'Disable "Sample stop" button
ToolbarEnable(2,1);					'Disable "Sample stop" button
ToolbarText("Press SAMPLE START to commence sampling");
return 1;
end;

proc DoToolbar()
ToolbarSet(0,"",Idle%);				'Call Idle%() whenever there is free time
ToolbarSet(1,"&Quit",Quit%);		'Set up toolbar buttons
ToolbarSet(2,"&Sample start", Start%);
ToolbarSet(3,"&Sample stop", Stop%);
ToolbarSet(4,"&New file", New%);
ToolbarEnable(3,0);					'Disable "Sample stop" button
Toolbar("Press SAMPLE START to commence sampling", 1023);'Wait here until quit
end;

func Quit%()							'If "Quit" is pressed
SampleStop();							'Stop sampling
return 0;								'leave toolbar
end;

func Start%()							'If "Start" is pressed
sTime := 0;                      'set start of anaylsis time
SampleStart();					 'Start sampling
ToolbarEnable(4,0);					'Disable "New file" button
ToolbarEnable(3,1);					'Enable "Sample stop" button
ToolbarEnable(2,0);					'Disable "Sample start" button
ToolbarEnable(1,0);					'Disable "Quit" button
ToolbarText("Press SAMPLE STOP to stop sampling");
return 1;								'Stay with toolbar
end;

func Stop%()							'If "Stop" is pressed
SampleStop();							'Stop sampling
if (data%>0) and                 'if no data in file, stop will close it
   (ViewKind(data%) <> 0) then data% := 0 endif;
ToolbarEnable(4,1);					'Enable "New file" button
ToolbarEnable(3,0);					'Disable "Sample stop" button
ToolbarEnable(1,1);					'Enable "Quit" button
ToolbarText("Press FILE NEW to capture more data");
return 1;								'Stay in toolbar
end;

func Idle%()							'Idle routine runs when system has time
var eTime;
if ViewKind(data%)=0 then        'This checks that the data view exists...
   View(data%);                  '...as this gives an error if it doesn't
   eTime:=Maxtime();
    
    if (eTime > sTime) then
        if MaxTime() > XLow()+5 then        'If current time has changed
            Draw((MaxTime()-off),range);    'Update the display accordingly
        endif;
        
      sTime:=eTime;
      endif;
   endif;
return 1;								'Stay in toolbar
end;

'You can find this code in the App() on-line help.
var gFloat%[20];              'global for floating window states
proc HideAll()
var i%;
gFloat%[0] := App(-2);        'number of windows
for i% := 1 to gFloat%[0] do  'hide all windows and save state
   gFloat%[i%] := View(App(i%)).WindowVisible(0);
   next;
end

proc RestoreAll()
var i%;
for i% := 1 to gFloat%[0] do  'restore hidden windows
   View(App(i%)).WindowVisible(gFloat%[i%]);
   next;
end




SampleWindow.s2s


